Quadriken 3d Hyperbolisches Paraboloid

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4, Juni 07   Update 30.06.07
Web: http://haftendorn.uni-lueneburg.de   www.mathematik-verstehen.de

##

Rückwärts aufgestellt, siehe Datei Konstruktion-rückwärts

```
p:=matrix([x,y,z]):pt:=linalg::transpose(p)
```

$\begin{pmatrix} x & y & z \end{pmatrix}$

Hyperbolisches Paraboloid

```
//kv1:=0: kv2:=-2: kv3:=3:
quadrik:=matrix([[5*x^2 - 26*x*y + 8*x*z - 88*x + 5*y^2
+ 8*y*z + 56*y - 4*z^2 + 34*z + 128]])
```

$$\left(5 \cdot x^2 - x \cdot y \cdot 26 + 8 \cdot x \cdot z - x \cdot 88 + 5 \cdot y^2 + 8 \cdot y \cdot z + 56 \cdot y - z^2 \cdot 4 + 34 \cdot z + 128 \right)$$

```
quadrikp:=plot::Implicit3d(quadrik[1]=0,x=-5..5,y=-10..2
,z=-5..5,FillColor=[0,1,0,1], Scaling=Constrained):
plot(quadrikp):
quadrikp:=plot::Implicit3d(quadrik[1]=0,x=-5..5,y=-10..2,z=-5..5,FillColor=[0,1
,0,0.8], Scaling=Constrained): //nochmal Transparent
```



A passend aufstellen

```
A:=matrix([[5,-13,4],[-13,5,4],[4,4,-4]]);
a:=matrix([-88,+56,34]): at:=linalg::transpose(a);
d:=128;
```

$$\begin{pmatrix} 5 & -13 & 4 \\ -13 & 5 & 4 \\ 4 & 4 & -4 \end{pmatrix}$$

$\begin{pmatrix} -88 & 56 & 34 \end{pmatrix}$

128

```
expand(pt*A*p+at*p+d);  //Probe, ob man A,a und d
```

```
expand(pt*A*p+at*p+d);  //Probe, ob man A,a und d
richtig hat
%-quadrik
```

$$\left(5 \cdot x^2 - x \cdot y \cdot 26 + 8 \cdot x \cdot z - x \cdot 88 + 5 \cdot y^2 + 8 \cdot y \cdot z + 56 \cdot y - z^2 \cdot 4 + 34 \cdot z + 128 \right)$$

$$(0)$$

hier muss 0 herauskommen

Hauptachtsentransformation

```
E3:=matrix([[1,0,0],[0,1,0],[0,0,1]])
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
evli:=linalg::eigenvectors(A) //Probe, was MuPAD liefert
```

$$\left[\left[0, 1, \left[\begin{pmatrix} \frac{1}{2} \\ \frac{1}{2} \\ 1 \end{pmatrix} \right] \right], \left[-12, 1, \left[\begin{pmatrix} -1 \\ -1 \\ 1 \end{pmatrix} \right] \right], \left[18, 1, \left[\begin{pmatrix} -1 \\ 1 \\ 0 \end{pmatrix} \right] \right] \right]$$

Eigenwerte und Eigenvektoren

```
ew1 :=evli[1][1];  ew2 :=evli[2][1]; ew3 :=evli[3][1];
ev1:=evli[1][3][1]:
ev2:=evli[2][3][1]:
ev3:=evli[3][3][1]:
```

$$0$$

$$-12$$

$$18$$

$$\begin{pmatrix} \frac{1}{2} \\ \frac{1}{2} \\ 1 \end{pmatrix}$$

$$\begin{pmatrix} -1 \\ -1 \\ 1 \end{pmatrix}$$

$$\begin{pmatrix} -1 \\ 1 \\ 0 \end{pmatrix}$$

```
ev1n:=linalg::normalize(ev1):
ev2n:=linalg::normalize(ev2):
ev3n:=linalg::normalize(ev3):
P:=ev1n.ev2n.ev3n: Pt:=linalg::transpose(P);
```


$$\begin{pmatrix} \frac{\sqrt{2} \cdot \sqrt{3}}{6} & \frac{\sqrt{2} \cdot \sqrt{3}}{6} & \frac{\sqrt{2} \cdot \sqrt{3}}{3} \\ -\frac{\sqrt{3}}{} & -\frac{\sqrt{3}}{} & \frac{\sqrt{3}}{} \end{pmatrix}$$

$$\begin{pmatrix} \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{3} \\ -\frac{\sqrt{3}}{3} & -\frac{\sqrt{3}}{3} & \frac{\sqrt{3}}{3} \\ -\frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \end{pmatrix}$$

Vektorschreibweise für die Abbildung $\vec{p} = P\,\vec{p}\,'$ und die Quadrikgleichungen, die sich durch Einsetzen ergeben:

$$Q: \quad \vec{p}^{\,T} A\,\vec{p} + \vec{a}^{\,T}\vec{p} + d = 0$$

$$Q': \quad \vec{p}\,'^{T} P^T A\, P\,\vec{p}\,' + \vec{a}^{\,T} P\vec{p}\,' + d = 0$$

$$Q': \quad \vec{p}\,'^{T} D_{EW}\,\vec{p}\,' + \vec{a}^{\,T} P\vec{p}\,' + d = 0$$

mit $D_{EW} = \begin{pmatrix} \lambda_1 & 0 \\ 0 & \lambda_2 \end{pmatrix}$

bzw. in 3D

```
quastrich:=Simplify(pt*Pt*A*P*p+at*P*p+d)
```

$$\left(22\cdot\sqrt{3}\cdot y - y^2\cdot 12 + 18\cdot z^2 + 72\cdot\sqrt{2}\cdot z + 6\cdot\sqrt{6}\cdot x + 128\right)$$

```
quastrichp:=plot::Implicit3d(quastrich[1],x=-5..5,y=-5..
4, z=-7..5,FillColor=[1,1,0,1]):
plot(quastrichp /*,Op, tp */, Scaling=Constrained):
```




```
quastrich
```


$$\left(22\cdot\sqrt{3}\cdot y - y^2\cdot 12 + 18\cdot z^2 + 72\cdot\sqrt{2}\cdot z + 6\cdot\sqrt{6}\cdot x + 128\right)$$

Die Arbeitsweise ist dieselbe wie bei der Herstellung der Scheitelform einer Parabel.
Hier durch Hinsehen:

```
//xterm:=hold(6*(x+5/3*sqrt(3))^2);expand(xterm);
yterm:=hold(-12*(y-11/12*sqrt(3))^2);expand(yterm);
zterm:=hold(18*(z+2*sqrt(2))^2);expand(zterm);
```

$$-12\cdot\left(y-\frac{11\cdot\sqrt{3}}{12}\right)^2$$

$$-y^2\cdot 12+22\cdot\sqrt{3}\cdot y-\frac{121}{4}$$

$$18\cdot(z+2\cdot\sqrt{2})^2$$

$$18\cdot z^2+72\cdot\sqrt{2}\cdot z+144$$

Also

```
128-(144-121/4)
```

$$\frac{57}{4}$$

```
xterm:=6*sqrt(6)*(x+57/(24*sqrt(6)));expand(xterm);
```

$$6\cdot\sqrt{6}\cdot\left(x+\frac{19\cdot\sqrt{6}}{48}\right)$$

$$6\cdot\sqrt{6}\cdot x+\frac{57}{4}$$

```
quastrichK:=xterm+yterm+zterm
```

$$18\cdot(z+2\cdot\sqrt{2})^2-\left(y-\frac{\sqrt{3}\cdot 11}{12}\right)^2\cdot 12+6\cdot\sqrt{6}\cdot\left(x+\frac{19\cdot\sqrt{6}}{48}\right)$$

```
quastrich-expand(quastrichK)
```

$$(\,0\,)$$

hier muss 0 herauskommen

Letzter Teil der Hauptachsentransformation ist die Translation t

```
t:=matrix([19*sqrt(6)/48, -11/12*sqrt(3),2*sqrt(2)]);
tt:=linalg::transpose(t):
```

$$\begin{pmatrix}\frac{19\cdot\sqrt{6}}{48}\\ -\frac{11\cdot\sqrt{3}}{12}\\ 2\cdot\sqrt{2}\end{pmatrix}$$

$\vec{p}\,''=\vec{p}\,'+\vec{t}$ also $\vec{p}\,'=\vec{p}\,''-\vec{t}$ Das ergibt:

```
quaH:=Simplify(expand((pt-tt)*Pt*A*P*(p-t)+at*P*(p-t)))+
```

```
quaH:=Simplify(expand((pt-tt)*Pt*A*P*(p-t)+at*P*(p-t)))+
6*sqrt(x)+128;
quadrikH:=ew1*x^2+ew2*y^2+ew3*z^2+6*sqrt(6)*x;
```

$$\left(18 \cdot z^2 - y^2 \cdot 12 + 6 \cdot \sqrt{x} + 6 \cdot \sqrt{6} \cdot x\right)$$

$$-y^2 \cdot 12 + 18 \cdot z^2 + 6 \cdot \sqrt{6} \cdot x$$

Angabe der Gleichung in der üblichen Form:

```
hold(x/sqrt(6)-y^2/3+z^2/2=0)
```

$$\frac{x}{\sqrt{6}} - \frac{y^2}{3} + \frac{z^2}{2} = 0$$

```
quadrikHp:=plot::Implicit3d(quadrikH,x=-2..2,y=-3..3,
z=-2..2,
Scaling=Constrained):
plot(%)
```



Bestimmung des ursprünglichen Mittelpunktes:

$$\vec{m}'' = \begin{pmatrix} 0 \\ 0 \end{pmatrix}, \quad \vec{m}' = -\vec{t}\,, \quad \vec{m} = P(-\vec{t})$$

nun in 3D

```
m:=Simplify(P*(-t))
```

$$\begin{pmatrix} \frac{11}{16} \\ -\frac{53}{16} \\ \frac{1}{8} \end{pmatrix}$$

Dieses ist der alte Mittelpunkt.

Bestimmung des Urbildes des rechten Hauptscheitels:

Bestimmung des Urbildes des rechten Hauptscheitels:

$$\vec{r}\,'' = \begin{pmatrix} r \\ 0 \end{pmatrix},\ \vec{r}\,' = \begin{pmatrix} r \\ 0 \end{pmatrix} - \vec{t},\ \vec{r} = P\begin{pmatrix} r \\ 0 \end{pmatrix} + \vec{m},\ r = r \cdot \overrightarrow{ev_1} + \vec{m}$$

Mit ev1 rechts ist hier der normierte 1. Eigenvektor gemeint. Alles jetzt in 3D

```
r:=matrix([-2/3*sqrt(6),2,2]): //Punkt auf letzem Bild
rurstr:=r-t;
rur:=float(P*rurstr);
```

$$\begin{pmatrix} -\frac{17\cdot\sqrt{6}}{16} \\ \frac{11\cdot\sqrt{3}}{12}+2 \\ 2-\sqrt{2}\cdot 2 \end{pmatrix}$$

$$\begin{pmatrix} -2.548080767 \\ -3.719653643 \\ -0.05363279495 \end{pmatrix}$$

```
r
```

$$\begin{pmatrix} -\frac{2\cdot\sqrt{6}}{3} \\ 2 \\ 2 \end{pmatrix}$$

```
mp:=plot::Point3d(m,PointSize=2, PointColor=[0,1,0]):
Op:=plot::Point3d([0,0,0], PointSize=2, PointColor=[1,0,0]):
rurp:=plot::Point3d(rur, PointSize=2, PointColor=[0,1,1]):
rursp:=plot::Point3d(rurstr, PointSize=2, PointColor=[1,0.5,0]):
rp:=plot::Point3d(r,PointSize=2, PointColor=[0,0,1]):
ev1urp:=plot::Arrow3d(m,m+3*ev1):
ev2urp:=plot::Arrow3d(m,m+3*ev2):
ev3urp:=plot::Arrow3d(m,m+3*ev3):
tp:=plot::Arrow3d(-t,[0,0,0],LineColor=[1,0,0]):
plot(quadrikp,mp,Op,rurp,tp,
     ev1urp,ev2urp,ev3urp,PointSize=2,Scaling=Constrained);
```






Test ob es eine Drehung allein oder gefolgt von Spiegelung ist. det=1 reine Drehung

```
linalg::det(Pt)
```

$$-1$$

```
plot(quadrikp,quastrichp,
quadrikHp,mp,Op,tp,rp,rurp,rursp,

ev1urp,ev2urp,ev3urp,Axes=Origin,Scaling=Constrained)
```



Bestimmung der Dreh-Spiegelachse

```
Pt
```

$$\begin{pmatrix} \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{3} \\ -\frac{\sqrt{3}}{3} & -\frac{\sqrt{3}}{3} & \frac{\sqrt{3}}{3} \\ -\frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \end{pmatrix}$$

```
evliP:=linalg::eigenvectors(Pt)
```

$$\left[\left[-1,\ 1,\ \left[\begin{pmatrix} -\dfrac{2\cdot\sqrt{2}-\sqrt{2}\cdot\sqrt{3}}{\sqrt{2}+2\cdot\sqrt{3}-2} \\ -\dfrac{\sqrt{2}\cdot\sqrt{3}+2}{\sqrt{2}+2\cdot\sqrt{3}-2} \\ 1 \end{pmatrix}\right]\right],\ \left[\frac{\sqrt{2}\cdot\sqrt{3}}{12}-\frac{\sqrt{3}}{6}-\frac{\sqrt{\frac{\sqrt{2}\cdot\sqrt{3}}{3}-\frac{\sqrt{2}}{3}-\frac{\sqrt{3}\cdot 2}{3}-}}{2}\right.\right.$$

```
float(evliP)
```

$$\left[\left[-1.0,\ 1.0,\ \left[\begin{pmatrix} -0.1316524976 \\ -1.54586606 \\ 1.0 \end{pmatrix}\right]\right],\ \left[0.4154490106-0.9096164684\cdot i,\ 1.0,\ \left[\begin{pmatrix} 0.0546 \\ 0.642 \end{pmatrix}\right.\right.\right.$$

```
dreh:=evliP[1][3][1]
```

$$\begin{pmatrix} -\dfrac{2\cdot\sqrt{2}-\sqrt{2}\cdot\sqrt{3}}{\sqrt{2}+2\cdot\sqrt{3}-2} \\ -\dfrac{\sqrt{2}\cdot\sqrt{3}+2}{\sqrt{2}+2\cdot\sqrt{3}-2} \\ 1 \end{pmatrix}$$

```
drehp:=plot::Arrow3d(5*dreh, LineColor=[1,0,1])
```

$$\text{plot::Arrow3d}\left([0,\ 0,\ 0],\ \left[-\frac{5\cdot(-\sqrt{2}\cdot\sqrt{3}+2\cdot\sqrt{2})}{\sqrt{2}+2\cdot\sqrt{3}-2},\ -\frac{5\cdot(\sqrt{2}\cdot\sqrt{3}+2)}{\sqrt{2}+2\cdot\sqrt{3}-2},\ 5\right]\right)$$

```
plot(quadrikp,quastrichp,
quadrikHp,mp,Op,tp,rp,rurp,rursp,

ev1urp,ev2urp,ev3urp,drehp,Axes=Origin,Scaling=Constrain
ed)
```





5
4
3
2
1
0
4
3
2
1
-2
-4
-4
-5
-6
-7
-8
-9